

File Input and Output

File I/O (Input and Output) is covered in chapter 9. We will only cover sequential text input and output, which is described in the first half of chapter 9.

Reading Data from a Text File

If you have stored data in a text file, using a program such as Notepad, you can also read it with VB.NET. Reading from a text file is useful to load the program with large amounts of data that would otherwise be tedious to type.

Data can be stored in files and accessed with a StreamReader object.

The steps to use the StreamReader object are as follows:

1. Execute a statement of the form

```
Dim readerVar As IO.StreamReader
```

A StreamReader is an object from the Input/Output class that can read a stream of characters coming from a disk or coming over the Internet. The Dim statement declares the variable readerVar to be of type StreamReader.

2. Execute a statement of the form

```
readerVar = IO.File.OpenText(filespec)
```

where filespec identifies the file to be read. This statement establishes a communications link between the computer and the disk drive for reading data from the disk. Data then can be input from the specified file and assigned to variables in the program. This assignment statement is said to “open the file for input.”

Just as with other variables, the declaration and assignment statements in Steps 1 and 2 can be combined into the single statement:

```
Dim readerVar As IO.StreamReader = IO.File.OpenText(filespec)
```

3. Read items of data in order, one at a time, from the file with the ReadLine method. Each datum is retrieved as a string. A statement of the form

```
strVar = readerVar.ReadLine
```

causes the program to look in the file for the next unread line of data and assign it to the variable strVar. The data can be assigned to a numeric variable if it is first converted to a numeric type with a statement such as

```
numVar = CDbI(readerVar.ReadLine)
```

Note: If all the data in a file have been read by ReadLine statements and another item is requested by a ReadLine statement, the item retrieved will have the value Nothing.

4. After the desired items have been read from the file, terminate the communications link set in Step 2 with the statement

```
readerVar.Close()
```

As an example, an estimate of the cost it takes to raise a child to age 18 is stored in the file C:\COSTS.TXT and it contains the following:

```
Housing
54000
Food
33000
Transportation
23000
Clothing
13000
Health
18000
Education
14000
Other
16000
```

This says that Housing costs \$54,000 while Food costs \$33,000 , etc.

Here is a program that reads in all the data and outputs the total cost:

```

Dim costFile As IO.StreamReader = IO.File.OpenText("C:\COSTS.TXT")
Dim strAll, strTemp As String
Dim intTotal As Integer

strAll = costFile.ReadLine()      ' Reads "Housing"
strTemp = costFile.ReadLine()     ' Reads 54000
intTotal += CInt(strTemp)         ' Conver to Integer
strAll = strAll & "," & costFile.ReadLine() ' Reads "Food"
intTotal += CInt(costFile.ReadLine()) ' Reads and converts 33000
strAll = strAll & "," & costFile.ReadLine() ' Reads "Transportation"
intTotal += CInt(costFile.ReadLine())
strAll = strAll & "," & costFile.ReadLine() ' Reads "Clothing"
intTotal += CInt(costFile.ReadLine())
strAll = strAll & "," & costFile.ReadLine() ' Reads "Health"
intTotal += CInt(costFile.ReadLine())
strAll = strAll & "," & costFile.ReadLine() ' Reads "Education"
intTotal += CInt(costFile.ReadLine())
strAll = strAll & "," & costFile.ReadLine() ' Reads "Other"
intTotal += CInt(costFile.ReadLine())
costFile.Close()                  ' Close the file

MessageBox.Show("For:" & strAll & " The total cost is :" & intTotal)

```

The output is:



Note that there is a lot of repeat code. We repeat the same two lines of code to read in the name and cost of an item. We could improve this code by using a loop that continued until we reached the end of the file. The function:

`filevar.Peek`

returns -1 when we have reached the end of the file, and the ASCII code of the next character otherwise. Here is the simpler code:

```

Dim costFile As IO.StreamReader = IO.File.OpenText("C:\COSTS.TXT")
Dim strAll, strTemp As String
Dim intTotal As Integer

strAll = ""
Do While (costFile.Peek <> -1)
    strAll &= " " & costFile.ReadLine() ' Reads next cost item
    strTemp = costfile.ReadLine()     ' Reads cost
    intTotal += CInt(strTemp)         ' Convert to Integer
Loop
costFile.Close()                     ' Close the file
MessageBox.Show("For:" & strAll & " The total cost is :" & intTotal)

```

Here is another example that reads in words from a dictionary file to solve this word puzzle: "Name a common word, besides tremendous, stupendous and horrendous, that ends in dous."

We can solve this problem by loading up a file of words and checking each one to see if:

- 1) It is more than 4 letters long
- 2) The word contains "dous" at the end

Assume we have a file of English words located at C:\WORDS.TXT

```
Dim wordFile As IO.StreamReader = IO.File.OpenText("C:\WORDS.TXT")
Dim s As String

Do While (wordFile.Peek <> -1)
    s = wordFile.ReadLine()
    If s.Length > 4 Then
        If s.EndsWith("dous") Then
            Console.WriteLine(s)
        End If
    End If
Loop
wordFile.Close() ' Close the file
```

Writing To Sequential Text Files

Here we'll cover just the very basics of how to write and create a text file from your program.

Creating a text file is a lot like opening a file for reading, except we open it for creation instead. The steps to create a new text file and write data to it are:

1. Create an `IO.StreamWriter` object:

```
Dim swriter As IO.StreamWriter = IO.File.CreateText(pathOnDisk)
```

This will create a blank file with the given pathname. If the file already exists, it will be destroyed! (There is a separate function, `IO.File.AppendText`, that will open a file but append to the end of it instead of destroying the file). Note the similarities to opening a file, which was `IO.File.OpenText(pathOnDisk)`

If we don't specify a full path then by default the file is placed in the current working directory (the bin directory of the project, if running from visual studio)

2. To place data in the file, use `WriteLine`, as we have used to write data to the console, except precede it by the Stream Writer variable:

```
swriter.WriteLine(data)
```

3. When you are done recording data to the file, close it:

```
swriter.Close()
```

The close statement breaks the link with the file on disk and frees up space in memory.

Here is an example that would "hello there" and the number 42 to disk:

```
Dim sw As IO.StreamWriter = IO.File.CreateText("c:\test.txt")
sw.WriteLine("hello there")
sw.WriteLine(40 + 2)
sw.Close()
```